What does Hilbert series of monomial ideals describe? I am trying to understand the point of Hilbert series of monomial ideals. I am confused because Macaulay has commands for hilbertSeries, hilbertPolynomial and hilbertFunction.
What does Hilbert series of monomial ideals describe?
 A: Let $I$ be a monomial ideal over a standard graded polynomial ring $S$ over a field $K$. Let standard monomials of $I$, $\mathrm{std}(I)$, are those monomials $m \in S \setminus I$. The Hilbert function $h_I: \mathbb{N} \to \mathbb{N}$ of $I$ (or more precisely, of the quotient ring $S/I$) is the map $d \mapsto |(\mathrm{std}(I))_d|$ where $d$ is any natural number and $(\mathrm{std}(I))_d$ is the set of all monomials of degree $d$ not in $I$. In other words the Hilbert function is the $K$-vector space dimension of the vector subspace generated by $(\mathrm{std}(I))_d$. It's a good exercise to draw some small examples in $K[x,y]$ and compute their Hilbert functions by hand.
While the first few values of the Hilbert function of $I$ can vary wildly, a classical theorem tells us that for large enough values of $d$ the Hilbert function eventually agrees with a polynomial. This polynomial is called the Hilbert polynomial of $I$. 
The Hilbert series $H_I$ of $I$ is the generating function of the Hilbert function, that is, $H_I = \sum_{d>0}h_I(d)x^d$. 
All three objects encode a wealth of information about the ideal in question. The book Combinatorial Commutative Algebra gives a gentle introduction to the topic.
The Hilbert series can be expressed as a rational function $H_I(x) = \frac{p(x)}{(1-x)^d}$ where $p(x)$ is a polynomial and $d$ is the (Krull) dimension of the quotient ring $S/I$. When $I$ is an artinian (monomial) ideal (i.e., some power of every variable is a minimal generator of $I$) there are only finitely many standard monomials. In this case the Krull dimension of $S/I$ is zero. So $H_I(x) = p(x)$ and $p(1)$ is the vector space dimension of $S/I$.
Example: Here's an example in Macaulay2. Consider the ideal $I = \langle a^6,a^3b,a^2b^4,b^5 \rangle$ in $\mathbb{Z}_{17}[a,b]$.
i1 : S = ZZ/17[a,b]; -- construct a polynomial ring

i2 : I = monomialIdeal "a6,a3b,a2b4,b5" -- make a monomial ideal

                     6   3    2 4   5
o2 = monomialIdeal (a , a b, a b , b )

i3 : apply (0..10, i -> hilbertFunction (i,I)) -- compute some values of the Hilbert function

o3 = (1, 2, 3, 4, 4, 3, 0, 0, 0, 0, 0)

Since $I$ is artinian the Hilbert function is zero for input values $\gg0$. So the Hilbert polynomial is exactly the zero polynomial. We can check this in Macaulay2.
i4 : hilbertPolynomial I

o4 = 0

o4 : ProjectiveHilbertPolynomial

Some care should be exercised when computing the Hilbert series of an ideal in Macaulay2. If we simply call hilbertSeries we obtain a rational function with denominator $(1-T)^v$ where $v$ is the number of variables in $S$.
i5 : hilbertSeries I

          4    5     6     7
     1 - T  - T  - 2T  + 3T
o5 = -----------------------
                  2
           (1 - T)

o5 : Expression of class Divide

To obtain the Hilbert series $p(x)/(1-T)^d$ where $d$ is the Krull dimension of $S/I$, use reduceHilbert.
i6 : reduceHilbert hilbertSeries I

                2     3     4     5
     1 + 2T + 3T  + 4T  + 4T  + 3T
o6 = ------------------------------
                  1

o6 : Expression of class Divide

Compare the coefficients in the numerator to the nonzero values of the Hilbert function in o3. Adding these coefficients shows that $S/I$ is a 17-dimensional $S$-vector space.
i7 : p = numerator oo; -- get the numerator of the reduced Hilbert series

i8 : sub (p, (ring p)_0 => 1) -- evaluate the numerator at $T=1$

o8 = 17

